

# Analysing DSGE models with Dynare

## Macroeconomic Theory

Markus Haavio

Bank of Finland

January 2010

- Dynare is preprocessor and a collection of Matlab (and GNU Octave) routines, which solves DSGE models
- The Dynare home page is <http://www.dynare.org/>
- The Dynare manual is available at <http://www.dynare.org/documentation-and-support/manual>
- and Dynare can be downloaded at <http://www.dynare.org/download>

# The structure of a typical Dynare file

- You tell Dynare what the variables, shocks and parameters (+parameter values) of your model are
- You write down the model equations
- You tell Dynare to find the steady state
- Dynare (log)linearizes the model around the steady state
- ... and solves the recursive equilibrium laws of motion of the linearized model
- Finally, the model is analyzed via impulse responses, stochastic simulations and moments

# The structure of a typical Dynare file

- var
  - you tell Dynare what the 'endogenous' variables of your model are
- varexo
  - you give the list of shocks in your model
- parameters
  - you give the list of parameters in the model
  - ... and then assign the parameter values
  - you may also find it useful to give some parameter transformations
    - e.g.  $\beta$  is a parameter, and you define steady state real interest rate  $rr = 1/\beta - 1$
    - more generally, you may give the steady state values of endogenous variables in terms of parameters

# The structure of a typical Dynare file: Writing down the model

- The model block starts with

```
model;
```

- ... and ends with

```
end;
```

- Between 'model' and 'end' you write down the necessary equations defining the dynamic equilibrium of the model
  - first-order conditions
  - constraints
  - (Remember: you need as many equations as there are variables)

- Time indices are given in parenthesis
  - $X_{t+1}$  is written  $X(+1)$ ,
  - $X_{t-1}$  is written  $X(-1)$
  - $X_t$  is written  $X$  (no time index needed for the current period)

# Things you should know when writing down the model

- In Dynare, the time index refers to the period when the value of the variable is determined
  - The value of the capital stock, which is used in production in period  $t$ , is determined in period  $t - 1$ 
    - In period  $t - 1$ , the agents decide how much to consume and invest, and they simultaneously determine the size of the capital stock that will be available in period  $t$
  - In a theory model we write the period  $t$  production function

$$Y_t = Z_t K_t^\alpha L_t^{1-\alpha}$$

... but in Dynare we write

$$Y = ZK(-1)^{alpha} L^{(1-alpha)}$$

# Things you should know when writing down the model

- Dynare linearizes the model around the steady state
- It does *not* log-linearize the model
- Remember the advantages of log-linearizations: We are dealing with percentage deviations from the steady state (or the balanced growth path)
  - Log-deviations (or percentage deviations) are easy to interpret (is a deviation 'small' or 'large' ?)
  - Similar measures are used in empirical examination of the data (e.g. applying HP filter to  $\log(\text{GDP})$ )
- A useful trick: Write down the model in terms of logarithmic transformations of the original variables
  - When Dynare linearizes the model in terms of the logarithmic transformations, it log-linearizes the model in terms of the original variables

# Things you should know when writing down the model

- Adopt the notation

$$ly = \log(Y), \quad lc = \log(C), \quad lk = \log(K) \quad \text{etc.}$$

- Note: For employment we use the notation

$$lh = \log(L)$$

where  $h$  refers to *hours* worked (the alternative  $ll$  just does not look very nice)

- Also remember the Dynare conventions pertaining to time indexation
- Then the resource constraint

$$Y_t = C_t + K_{t+1} - (1 - \delta) K_t$$

can be written as

$$\exp(ly) = \exp(lc) + \exp(lk) - (1 - \text{delta}) * \exp(lk(-1))$$

# Things you should know when writing down the model

- The expectation operator  $E_t$  is not used in Dynare code.
  - Dynare 'knows' when one has to take expectations, we do not have to tell this explicitly.
- Thus the consumption Euler equation

$$E_t \left[ \frac{C_{t+1}}{\beta C_t} \right] = E_t \left[ \alpha Z_{t+1} \left( \frac{K_{t+1}}{L_t} \right)^{\alpha-1} + 1 - \delta \right]$$

or

$$E_t \left[ \frac{C_{t+1}}{\beta C_t} \right] = E_t \left[ \alpha \frac{Y_{t+1}}{K_{t+1}} + 1 - \delta \right]$$

is written as

$$\frac{1}{\text{beta}} * \exp(\text{lc}(+1) - \text{lc}) = \text{alfa} * (\exp(\text{y}(+1) - \text{lk}) + 1 - \delta)$$

- Notice:  $lh = \log(L)$  ( $h$  refers to *hours worked*)

# Finding the steady state

- It is often useful to give Dynare an initial guess of the steady state values, after the command

```
initval;
```

- You tell Dynare to find the steady state with the command

```
steady;
```

# Characterizing the shock structure of the model

- shocks
  - You tell Dynare that you will now characterize the shock structure of the model
- var e
  - Here we want to shock only one 'variable' (the TFP shock e)
- stderr 0.007;
  - The standard error of the TFP shock is 0.007
  - If there are several shocks, say e\_z and e\_c, you have to be more careful
    - `stderr e_z 0.007; stderr e_c 0.002;`
  - More generally, if there are several shocks, you may want to give the variance-covariance matrix of the shocks; see Dynare Manual
- end;
  - You tell Dynare that the characterization of the shock structure ends

# Analyzing the model via impulse responses, stochastic simulations and moments

- `stoch_simul(order=1,irf=100) ly lh lc li lk z;`
  - `order=1` means that Dynare takes a first order Taylor approximation around the steady state (`order=2` => second order approximation, `order=3` => third order approximation)
  - `irf=100` means that you want Dynare to compute the impulse responses for 100 periods
  - Dynare also computes the moments of the endogenous variables
    - mean, standard deviation, variance, skewness, kurtosis
    - (contemporaneous) covariance matrix
    - autocorrelations (default: up to the 5th order)
  - Here the list of variables, we want to Dynare to analyze is `ly lh lc li lk z`
    - Notice that  $z \equiv \log(Z_t) - \log(\bar{Z}) = \log(Z_t)$ , since  $\bar{Z} = 0$  and  $\log(\bar{Z}) = 0$  (a useful normalization).

# Analyzing the model via impulse responses, stochastic simulations and moments

Alternatively we can write:

- `stoch_simul(order=1,periods=1000,irf=100) ly lh lk lc lw z;`
- Now Dynare simulates the model for 1000 periods, and computes the moments based on the simulated data

# The structure of the code in a nutshell

```
var (list of endogenous variables );  
varexo (list of shocks );  
parameters (parameters + parameter values + transformations);  
  
model;  
model equations;  
end;  
  
initval; (initial guesses for computing the steady state)  
steady; (compute the steady state)  
  
shocks; (the shock structure of the model )  
var (what variables are shocked);  
stderr (the standard error of the shocks);  
end;  
stoch_simul(order=1,irf=100) ly lh lc li lk z; (analyzing the model)
```

# Hansen's RBC model: key equations

$$1 = \beta E_t \left[ \left( \frac{C_t}{C_{t+1}} \right) \left( \alpha Z_t \left( \frac{K_{t+1}}{L_{t+1}} \right)^{\alpha-1} + 1 - \delta \right) \right]$$

$$\psi C_t = (1 - \alpha) Z_t \left( \frac{K_t}{L_t} \right)^\alpha$$

$$Y_t = C_t + K_{t+1} - (1 - \delta) K_t$$

$$I = Y_t - C_t$$

$$W_t = (1 - \alpha) \frac{Y_t}{L_t}$$

$$\log Z_t = (1 - \phi) \log \bar{Z} + \phi \log Z_{t-1} + \varepsilon_t, \varepsilon_t \text{ is iid}$$

# Dynare file: Hansen's RBC model

```
// following are the 'endogenous' variables:
```

```
var ly, lh, lc, li, lw, lk, z, ;
```

```
// following are the shocks
```

```
varexo e;
```

```
parameters beta, fi, alpha, delta, psi;
```

```
alpha = 0.33;
```

```
fi = 0.74;
```

```
beta = 0.99;
```

```
delta = 0.024;
```

```
psi =0.262;
```

```
// Steady state expressed in terms of parameters
rr=1/beta-1;
zss=1;
css=(1-alpha)*(zss/psi)*(alpha*zss/(rr+delta))^(alpha/(1-alpha));
kss=css/((rr+delta)/alpha-delta);
hss=(1/psi)*(1-alpha*rr/(rr+(1-alpha)*delta));
yss= zss*kss^(alpha)*hss^(1-alpha);
iss=delta*kss;
wss=(1-alpha)*yss/hss;
```

# Dynare file: Hansen's RBC model

```
model;
```

```
(1/beta)*exp(lc(+1)-lc)= alpha*exp(ly(+1)-lk) + 1 - delta;  
//Euler equation
```

```
psi*exp(lc)= (1-alpha)*exp(ly-lh); //Labour market equilibrium
```

```
exp(ly) = exp(lc) + exp(lk) - (1-delta)*exp(lk(-1)); //Resource constraint
```

```
exp(ly) = exp(z+alpha*lk(-1)+(1-alpha)*lh); //Production function
```

```
exp(li)=exp(ly)-exp(lc); // Investment
```

```
exp(lw)=(1-alpha)*exp(ly-lh); //Real wage
```

```
z = fi*z(-1) + e; //Law of motion of TFP
```

```
end;
```

```
//Initial guesses for the computation of steady state
```

```
initval;  
lc=log(css);  
lh=log(hss);  
lk=log(kss);  
ly=log(yss);  
li=log(iss);  
lw=log(wss);  
z=zss;  
end;
```

```
steady;
```

```
shocks;  
var e;  
stderr 0.007.;  
end;  
  
stoch_simul(order=1,irf=100) ly lh lc li lk z;
```

# Saving and running the Dynare file

- You save the file as [Name of the file].mod
  - For example rbc.mod
- You run the model by writing to the Matlab command window

```
dynare rbc.mod
```

- You may also run the Dynare file from a separate m.-file (this is practical if you, say, want to solve the model many times, with different parameter values)

## Output includes

- Policy and transition functions of the (log)linearized model
- Moments of the endogenous variables
  - mean, variance, standard deviation, skewness, kurtosis
  - matrix of contemporaneous correlations
  - coefficients of autocorrelation
- Impulse responses
- If if you have included the periods option in

```
stoch_simul(order=1,periods=1000,irf=100) ly lh lc li lk z;
```

the output also includes the results (time paths of endogenous variables) from the stochastic simulation.

- The output is shown on the screen, and in separate figures (impulse responses).
- Output is also stored in a separate structure, called `oo_`
- The structure `oo_` contains (for example)
  - The steady state (`oo_.steady_state`)
  - The variance-covariance matrix (`oo_.var`)
  - The autocorrelations (`oo_.autocorr`)
  - The impulse responses (`oo_.irfs`)
  - The coefficients of the policy and transition functions (`oo_.dr`)
  - Results (time paths of endogenous variables) from stochastic simulations (`oo_.endo_simul`)

# Output: Policy and transition functions

	$ly$	$lh$	$lc$	$li$	$lk$	$z$
Constant	0.019	-1.1	-0.24	-1.44	2.29	0
$lk(-1)$	-0.0075	-0.50	0.50	-1.67	0.94	0
$z(-1)$	1.87	1.68	0.18	7.4	0.18	0.74
$e$	2.52	2.28	0.25	10.0	0.24	1

- Transition functions: how the period  $t$  values of the state variables ( $lk$  and  $z$ ) depend on the period  $t - 1$  values of the state variables, and the shock
- Policy functions: how the period  $t$  values of the other variables depend on the period  $t - 1$  values of the state variables, and the shock

# Output: Policy and transition functions

	$ly$	$lh$	$lc$	$li$	$lk$	$z$
Constant	0.019	-1.1	-0.24	-1.44	2.29	0
$lk(-1)$	-0.0075	-0.50	0.50	-1.67	0.94	0
$z(-1)$	1.87	1.68	0.18	7.4	0.18	0.74
$e$	2.52	2.28	0.25	10.0	0.24	1

- Rows: period  $t$  values of the variables
- Columns: period  $t - 1$  values of the state variables + the shock ( $e$ )
- The term 'constant' is the steady state (of log transformation)
  - E.g.  $\bar{lh} = -1.1$  and thus  $\bar{L} = \exp(-1.1) = 0.33$

# Output: Policy and transition functions

	$ly$	$lh$	$lc$	$li$	$lk$	$z$
Constant	0.019	-1.1	-0.24	-1.44	2.29	0
$lk(-1)$	-0.0075	-0.50	0.50	-1.67	0.94	0
$z(-1)$	1.87	1.68	0.18	7.4	0.18	0.74
$e$	2.52	2.28	0.25	10.0	0.24	1

- The entries of the table give the coefficients of the (log)linearized policy and transition functions. For example (the first column)

$$ly = \underbrace{0.019}_{\bar{ly}} - 0.0075 * (lk(-1) - \bar{lk}) + 1.87 * z(-1) + 2.52 * e$$

## Output: Connection to our earlier results (Lecture 2)

$$ly = \underbrace{0.019}_{\bar{ly}} - 0.0075 * (lk(-1) - \bar{lk}) + 1.87 * z(-1) + 2.52 * e \quad (1)$$

- How to write (1) in terms of log deviations  $y_t$ ,  $k_t$  and  $z_t$ ?
- Remember that

$$\begin{aligned} ly - \bar{ly} &= \log(Y_t) - \log(\bar{Y}) = y_t; \\ lk(-1) - \bar{lk} &= \log(K_t) - \log(\bar{K}) = k_t \end{aligned}$$

- Also notice that

$$2.52 * z = \underbrace{2.52 * 0.74}_{1.87}^{\phi} * z(-1) + 2.52 * e$$

- Thus (1) can be rewritten as

$$y_t = -0.0075 * k_t + 2.52 * z_t$$

## Output: Connection to our earlier results (Lecture 2)

	$ly$	$lh$	$lc$	$li$	$lk$	$z$
Constant	0.019	-1.1	-0.24	-1.44	2.29	0
$lk(-1)$	-0.0075	-0.50	0.50	-1.67	0.94	0
$z(-1)$	1.87	1.68	0.18	7.4	0.18	0.74
$e$	2.52	2.28	0.25	10.0	0.24	1

- More generally, in terms of the notation presented in Lecture 2, the transition and policy functions can be re-expressed as follows

$$k_{t+1} = 0.94 * k_t + 0.24 * z_t; \quad c_t = 0.50 * k_t + 0.25 * z_t$$

$$h_t = -0.50 * k_t + 2.28 * z_t; \quad l_t = -1.67 * k_t + 10.0 * z_t$$

- Thus we have the correspondence:

In the theory model	In Dynare
The coefficient of $k_t$	= The coefficient of $lk(-1)$
The coefficient of $z_t$	= The coefficient of $e$